TRADESTATION: REVERSE ENGINEERING RSI 

Giorgos Siligardos' article in this issue, "Reverse Engineering RSI," describes the calculation and use of a translation process to plot the "next bar" price required to produce any given RSI value. Here, we present the equivalent EasyLanguage code, named "RevEngRSI-Flat," and a modest variation plotting the equivalent price to a diagonal Rsi trend (RevEngRSI-Diag). These two new EasyLanguage indicators are designed to be used in tandem with the built-in TradeStation Rsi indicator. 

FIGURE 2: TRADESTATION, REVERSE RSI, WEEKLY SPY. Here's a sample TradeStation chart of the weekly SPY displaying the RevEngRSI-Flat and the RevEngRSI-Diag indicators. Use these indicators with the built-in TradeStation Rsi indicator.


To plot the EasyLanguage indicators, start with a chart of any given security. My example is a weekly chart of the Spy (Figure 2). Then: 
  

1.  Insert the TradeStation Rsi indicator into subplot 2. 
2.  Select the drawing tool and create a horizontal line that seems to capture the RSI's range. While you are drawing the line, you should be able to see the RSI value displayed. When you have the right line, write down the RSI value. It will be used in the next step. 
3.  Insert the RevEngRsi-Flat indicator and immediately go to format "inputs." Make sure the RSI length matches the plotted Rsi length. Change the "RsiValue" to the value captured from your horizontal line drawn in step 2. 
4.  Repeat steps 2 and 3 as many times as you like. Each time, you will get a new price plot representing the "next bar" price required to reach the desired RSI level.


If you find the RSI seems to bounce off a diagonal trend, you can use RevEngRSI-Diag to plot the price forecast. After you have gone through steps 1 through 4: 
  

5.  Select the "TrendLine" tool and draw a trendline from the first extreme to the last extreme on your RSI subgraph. 
6.  Select and hold the resulting trendline. You will get a display describing the trendline's start and endpoints (date, time and Rsi values). Write these down. 
7.  Insert the indicator RevEngRSI-Diag and immediately format "inputs." Make sure the RSI length is the same as your standard Rsi. Enter the start and ending values. 
8.  Enter the number of bars you expect the trendline to be valid after the entered trendline endpoint. 
9.  Repeat steps 5 through 8 as many times as you like.

Indicator: RevEngRSI-Flat

inputs:

 RSIValue( 50 ),

 WilderTimePeriod( 14 ) ;

variables:

 ExpPer( 2 * WilderTimePeriod  - 1 ),

 AUC( 0 ),

 ADC( 0 ),

 X( 0 ),

 RevEngRSI( 0 ) ;

AUC = XAverage( IFF( C > C[1], C - C[1], 0), ExpPer ) ;

ADC = XAverage( IFF( C[1] > C, C[1] - C, 0), ExpPer ) ;

X = (WilderTimePeriod - 1 ) * ( ADC * RSIValue

 / ( 100 - RSIValue ) - AUC ) ;

RevEngRSI = IFF( X >= 0, C + X, C + X *

 (100-RSIValue)/RSIValue) ;

Plot1[-1]( RevEngRSI ) ;

Indicator: RevEngRSI-Diag

inputs:

 WilderTimePeriod( 14 ),

 RSI_TL_StartDate( 1020719 ), {default Jul 19, 2002}

 RSI_TL_StartTime( 1515 ),

 RSI_TL_StartValue( 21.93 ),

 RSI_TL_EndDate( 1030307 ), {default Mar 7, 2003}

 RSI_TL_EndTime ( 1515 ),

 RSI_TL_EndValue( 39.12 ),

  AdditionalBars( 20 ) ;

variables:

 ExpPer( 2 * WilderTimePeriod - 1 ),

 AUC( 0 ),

 ADC( 0 ),

 X( 0 ),

 RevEngRSI( 0 ),

 Count( 0 ),

 TL_RSIValue( 0 ),

 TL_RevEngRSI( 0 ),

 Slope( 0 ),

 OldTL_RevEngRSI( 0 ),

 MaxCount( 0 ) ;

AUC = XAverage( IFF( C > C[1], C - C[1], 0), ExpPer ) ;

ADC = XAverage( IFF( C[1] > C, C[1] - C, 0), ExpPer ) ;

if Date  = RSI_TL_EndDate and Time = RSI_TL_EndTime then

 begin

 Count = 0 ;

 while Date[Count] >= RSI_TL_StartDate and Count <100

  begin

  Count = Count + 1 ;

  end;

 Count = Count - 1 ;

 MaxCount = Count + AdditionalBars ;

 Slope = ( RSI_TL_EndValue - RSI_TL_StartValue )

  / Count ;

 TL_RSIValue = RSI_TL_StartValue ;

 X = ( WilderTimePeriod - 1 )

  * ( ADC[Count] * TL_RSIValue

  / ( 100 - TL_RSIValue ) - AUC[Count] ) ;

 TL_RevEngRSI = IFF( X >= 0, C[Count] + X,

  C[Count] + X * ( 100 - TL_RSIValue )

  / TL_RSIValue ) ;

 For Value1 = 1 to Count

  begin

  OldTL_RevEngRSI = TL_RevEngRSI ;

  TL_RSIValue = Slope * Value1

   + RSI_TL_StartValue ;

  Value2 = Count - Value1 ;

  X = ( WilderTimePeriod - 1 )

   * ( ADC[Value2] * TL_RSIValue

   / ( 100 - TL_RSIValue ) - AUC[Value2] ) ;

  TL_RevEngRSI = IFF( X >= 0,

   C[Value2] + X,

   C[value2]+ X * ( 100 - TL_RSIValue )

   / TL_RSIValue ) ;

  Value99 = TL_New( Date[Value2], Time[Value2],

   OldTL_RevEngRSI, Date[Value2-1],

   Time[Value2-1], TL_RevEngRSI ) ;

  end ;

 end

else if ( Date > RSI_TL_EndDate )

 or ( Date = RSI_TL_EndDate and Time > RSI_TL_EndTime)

 and Count < MaxCount

then

 begin

 Count = Count + 1 ;

 OldTL_RevEngRSI = TL_RevEngRSI ;

 TL_RSIValue = Slope * Value1 + RSI_TL_StartValue ;

 Value2 = Count - Value1 ;

 X = ( WilderTimePeriod - 1 ) * ( ADC * TL_RSIValue

  / ( 100 - TL_RSIValue ) - AUC ) ;

 TL_RevEngRSI = IFF( X >= 0, C + X,

  C + X * ( 100 - TL_RSIValue ) / TL_RSIValue ) ;

 Value99 = TL_New( Date[1], Time[1],

  OldTL_RevEngRSI[1], Date, Time,OldTL_RevEngRSI);

 end ;


This indicator code will be available for download from the EasyLanguage Exchange on TradeStationWorld.com. Look for the file "RegEngRSI.eld". 

--Mark Mills 
MarkM@TSSec at www.TradeStationWorld.com 
EasyLanguage Questions Forum 
TradeStation Securities, Inc. 
A subsidiary of TradeStation Group, Inc.

The following program for MetaStock computes the closing
price at which the Rsi of the previous day will attain a
specific level for the next day.

{Reverse Engineered RSI}

value:=Input(“RSI value”, 1, 100,50);
WildPer:=Input(*Wilder Time Peri-
ods",1, 100,14);
ExpPer:=2"WildPer-1;

{Average Up Close}

AUC:=Mov( If(C>Ref(C,-1), C-
Ref(C,-1), 0), ExpPer, E);
{Average Down Close}

ADC:=Mov( If(Ref(C,-1)>C, Ref(C,-
1)-C, 0), ExpPer, E);

x:=(WildPer-1)*(ADC"value/(100-
value)-AUC);

RevEngRSI:=If(x>=0, C+x, C+x*(100-
value)/value);

RevEngRSI

Note: The smoothing method of k periods provided by
Wilder for smoothing the up and down closes was indeed a
(2k—1)-period exponential moving average of UC and DC,
respectively. For example, Wilder calculates the AUC as:

(k= 1)fAUC, iy *+ U,

AUC,

oy = ey

l"ucw +=Laue

which is by definition the (2k—1)-period exponential moving
average of UC. —GS








